Computing the prime ideals in the ring of complex polynomials in two variables

rapha • 1 May 2025

Long winded title, couldn't figure out how to make $\mathbb{C}[x, y]$ appear in the title. That's the ring for which I'm going to compute all of the prime ideals. That's useful to know because that computation gives the points of the affine scheme $\mathbb{A}^2_{\mathbb{C}}$, the "complex affine 2-space".

First of all, $\mathbb{C}[x, y]$ is an integral domain, hence (0) is a prime ideal. Also, this ring is a unique factorization domain (UFD), so any irreducible polynomial $p(x, y)$ is also prime, whence (p) is a prime ideal. Finally, for any two $\alpha, \beta \in \mathbb{C}$, the ideal $(x - \alpha, y - \beta)$ is maximal so a fortiori it's prime. Intuitively, the zero (0) is a "two-dimensional point", the (p)'s are "one-dimensional points", and the $(x - \alpha, y - \beta)$ should be thought at "standard, zero-dimensional points" (i.e. *closed points*).

I claim that there are no prime ideals other than those I just enumerated. First, the easy case. Suppose $\mathfrak{p} \neq 0$ is a *principal* prime ideal, i.e. it has a single generator f and $\mathfrak{p} = (f)$. Because $\mathfrak{p}$ is not the zero ideal, the element f is prime, and so it is an irreducible polynomial.

Now the harder case: suppose $\mathfrak{p}$ is a prime ideal that is *not* principal. We want to show it is of the form $(x - \alpha, y - \beta)$ for some complex numbers α and β. A daunting task! In fact, not so daunting. Here's a fact that will get us started: there necessarily exists two relatively prime polynomials f and g in $\mathfrak{p}$. "Relatively prime" means that any element that divides both f and g must be a unit, i.e. a complex number. The proof idea for this factoid is the following: choose F to be any non-unit, non-zero polynomial in $\mathfrak{p}$. Since our ring is a UFD, we may decompose F into a product of irreducibles, one of which will be in $\mathfrak{p}$. Define f to be this irreducible factor lying in $\mathfrak{p}$. Because $\mathfrak{p}$ is not principal, there must be some other polynomial g in the non-empty set $\mathfrak{p} \setminus (f)$. Then f and g are relatively prime. If my proof sketch is not convincing your, here's a proof of it in Lean 4, leveraging Mathlib; feel free to judge my programming skills.

```
import Mathlib.Algebra.Group.Units.Basic
import Mathlib.Algebra.Polynomial.Basic
import Mathlib.Algebra.Polynomial.Bivariate
```

```
import Mathlib.Algebra.Prime.Defs
import Mathlib.Data.Complex.Basic
import Mathlib.RingTheory.Ideal.Basic
import Mathlib.RingTheory.Ideal.IsPrincipal
import
        Mathlib.RingTheory.Polynomial.UniqueFactorizat
        ion
import
        Mathlib.RingTheory.UniqueFactorizationDomain.B
        asic

open Polynomial Bivariate Submodule
        UniqueFactorizationMonoid

abbrev A := ℂ[X][Y]

lemma exists_rel_prime {I : Ideal A} (hI : I.IsPrime)
        (hI₂ : ¬I.IsPrincipal) : ∃ f g, f ∈ I ∧ g ∈ I
        ∧ IsRelPrime f g := by
  have hI₃ : I ≠ ⊥ := by
    intro h
    rw [<- Ideal.span_zero] at h
    exact hI₂ <| IsPrincipal.mk ⟨_, h⟩
  obtain ⟨f : A, hf : f ∈ I, hf₂ : Prime f⟩ :=
        hI.exists_mem_prime_of_ne_bot hI₃
  have compl_nonempty : ((I : Set A) \ (Ideal.span
        { f } : Set A)).Nonempty := by
    rw [Set.nonempty_iff_ne_empty]
    intro h
    rw [Set.diff_eq_empty] at h ; norm_cast at h
    replace h : I = Ideal.span { f } := by
      apply le_antisymm h
      rwa [Ideal.span_le, Set.singleton_subset_iff]
    exact hI₂ <| IsPrincipal.mk ⟨_, h⟩
  let g : A := compl_nonempty.some
  obtain ⟨hg : g ∈ I, hg₂ : g ∉ Ideal.span { f }⟩ :=
        compl_nonempty.some_mem
  use f, g, hf, hg
  intro d hdf hdg
  replace hf₂ : Irreducible f :=
        irreducible_iff_prime.mpr hf₂
  by_contra h
  rw [dvd_iff_exists_eq_mul_right] at hdf
  rcases hdf with ⟨c, hc⟩
  rcases Irreducible.isUnit_or_isUnit hf₂ hc with
        d_unit | c_unit
  . contradiction
```

```
    have hfg : f ∣ g := by
      refine dvd_trans ?_ hdg
      rw [dvd_iff_exists_eq_mul_right, hc]
      obtain ⟨c_inv, hc_inv⟩ := c_unit.exists_right_inv
      use c_inv
      rw [mul_assoc, hc_inv, mul_one]
    rw [<- Ideal.mem_span_singleton] at hfg
    exact hg₂ hfg
```

Let's continue. We have two polynomials f and g which are relatively prime members of $\mathfrak{p}$, and these two polynomials continue to be relatively prime when interpreted in the larger ring $\mathbb{C}(x)[y]$, by some other post I made. It is advantageous to work in this larger setting because $\mathbb{C}(x)[y]$ is an Euclidean domain, hence we may compute the greatest common divisor (GCD) of f and g in it. Moreover, Bézout's Identity is verified in this setting: we have the equality of ideals $(f, g) = (1)$, since the GCD of f and g is a unit by relative primality. In particular, there exists two polynomials $a, b \in \mathbb{C}(x)[y]$ such that $af + bg = 1$. In this expression, a and b may contain monomials terms for which the coefficient is a fraction with some polynomial in x as the denominator. We may rewrite the left hand side to put all these monomials on a common denominator, and multiply both sides by an appropriate polynomial $h(x)$ to lift that equation to $afh + bgh = h$ in $\mathbb{C}[x, y]$.

The previous equation shows that $h(x)$ lies in the ideal (f, g) generated by f and g. In particular, h lies in $\mathfrak{p}$. Because h is a non-constant, non-zero polynomial in a single variable with coefficients in $\mathbb{C}$, it admits at least one root $\alpha \in \mathbb{C}$ such that $x - \alpha \in \mathfrak{p}$.

We can repeat the same argument but in $\mathbb{C}(y)[x]$ to find some polynomial $k(y) \in \mathfrak{p}$. Therefore $(x - \alpha, y - \beta) \leq \mathfrak{p}$. But since $(x - \alpha, y - \beta)$ is a maximal ideal, we must have equality! ■